BINARY Vol. 3 © 1991 Academic Press Limited

A FORTRAN Subroutine for the Determination of Parameter Confidence Limits in Non-linear Models

J.R. Lobry, L. Rosso and J.P. Flandrois[1]

Laboratoire de Biométrie, CNRS URA 243, Université Claude Bernard, 43 Boulevard du 11 Novembre 69622 Villeurbanne, France
Electronic mail: DYNBACT@FRCISM51 (EARN)
[1]Faculté de Médecine Lyon-Sud, Laboratoire de Bactériologie, CNRS URA 243, Hôtel Dieu 69288 Lyon Cedex 02, France

Received: 17 January 1991; revised 21 March1991

A FORTRAN subroutine to determine confidence limits of parameter values in non-linear models was written. This subroutine writes into an output file a list of point coordinates, in the parameter space, for which the sum of squared residuals between theoretical and observed values equals the threshold value that defines the confidence region. This routine was used to determine parameter confidence limits of two different models describing the relationship between the specific growth rate of a microbial population as a function of temperature. Both models have four parameters and fit data almost equally well, although the structural correlation between parameters they yield is considerably different.

Introduction

This introduction briefly presents the two models used throughout this paper to illustrate the kind of results that may be obtained with the program given in the appendix. Both models give a relationship between the specific growth rate of a microbial population and the temperature. Their parameter values may then be used to summarize the behaviour of a strain as a function of temperature. Two different models were used not for comparison, but to illustrate two extreme conditions that may arise when determining confidence limits for parameters.

The first model (Ratkowsky *et al.*, 1983), which is an extension of a previously published one (Ratkowsky *et al.*, 1982), expresses the specific growth rate μ(h^{-1}) as a function of the absolute temperature T(K),

$$\mu = [b\,(T - T_{min})(1 - e^{c\,(T - T_{max})})]^2$$

where T_{min} is the minimum temperature (K), T_{max} the maximum temperature (K), b the regression coefficient of the root of the specific growth rate versus absolute temperature below the optimum temperature (K^{-1}h$^{-0.5}$), and c a parameter which enables the model to be flexible enough to fit data at temperatures above the optimum temperature (K^{-1}).

The second model is new. It will be referred to here as the 'Cardinal Temperature Model' as the three cardinal temperatures are explicitly present in its expression,

$$\mu = \mu_{opt}\left[1 - \frac{(T - T_{opt})^2}{(T - T_{opt})^2 + T\,(T_{max} + T_{min} - T) - T_{max}.T_{min}}\right]$$

where μ is the specific growth rate (h^{-1}), T the absolute temperature (K), T_{min} the minimum temperature (K), T_{opt} the optimum temperature (K), T_{max} the maximum temperature (K), and μ_{opt} the specific growth rate at the optimum temperature (h^{-1}).

Both models cover the full biokinetic temperature range from T_{min} to T_{max} and have the same number of parameters. Figure 1 summarizes the effects of the four parameters of each model on the curve pattern.

Material and methods

Data

The data set was deduced from Figure 2 in the paper submitted by Ratkowsky *et al.* (1983). This figure gives the square root of the specific growth rate of *Escherichia coli* as a function of temperature. There are 15 available points in this data set, so that the ratio of the number of points over the number of parameters to be estimated (15/4) is greater than three, which seems a reasonable a priori. The numerical values used here are included in the program listing to enable reproduction of the results.

Data Processing

All computations were carried out on a Apple Macintosh IIfx computer with 8 Mbyte RAM, 160 Mbyte hard disk and version 6.05 of the Macintosh operating system. Programs were written in FORTRAN using Language Systems *FORTRAN Compiler 2.0* (Language Systems Corporation, 1989) under MPW 3.0 (Apple Computer Inc., 1988). Parameter estimations and model simulations were made with *Mathematica* (Wolfram Research Inc., 1988). Graphics were drawn with *Statview II* (Abacus Concepts Inc., 1988) and *SuperPaint* (Silicon Beach Software Inc., 1989).

Parameter Estimation

The Sum of Squared Residuals (SSR) was used to characterize the goodness of fit. The SSR is the sum of all the points of the squared difference between observed μ_i and theoretical μ^*_i specific growth rate values. The SSR is therefore a function of the set of parameter values Θ used to obtain μ^*_i,

$$SSR\,(\Theta) = \sum_{i=1}^{n} (\mu_i - \mu^*_i)^2$$

0226-304X/91/030086+08 $03.00/0



Figure 1 Effect of the four parameters of the Ratkowsky and of the Cardinal Temperature models. Parameters were varied one by one, giving four different plots of the specific growth rate (h^{-1}) versus temperature (K). The values of the varying parameter are shown near each curve. The three remaining parameters were always set to their default values which were: $b = 0.3$, $c = 0.035$, $T_{min} = 276$, $T_{max} = 322$ for the Ratkowsky model; and $\mu_{opt} = 1.4$, $T_{min} = 286$, $T_{opt} = 313$, $T_{max} = 323$ for the Cardinal Temperature Model.

The smaller the SSR, the better the fit. The set of parameter values $\hat{\Theta}$ with the lowest SSR was found using the built-in 'FindMinimum' subroutine of *Mathematica* (this routine is based on the usual steepest descent gradient).

Confidence Region

Confidence regions were determined using the program given in the appendix and described below.

When the SSR is used to characterize the goodness of fit, a 1-α confidence region for parameters is defined according to Beale (1960) by the set of parameter values Θ such that the SSR is less than a threshold,

$$\left\{ \Theta : SSR\ (\Theta) \leq SSR\ (\hat{\Theta})(1 + \frac{p}{n-p} F^{\alpha}_{p\ ,n\ -p}) \right\}$$

where p is the number of parameters and n the number of points. This equation defines a confidence region for parameter values in the parameter space. As the number of parameters is not limited, this confidence region can be described as a hypervolume bound by a hypersurface. The extent of this hypersurface must then be estimated to obtain confidence limits for parameter values.

One solution consists of computing a list of point coordinates belonging to this hypersurface. The number of points required to obtain a good representation of the hypersurface is difficult to determine as it depends on the regularity of the hypersurface. A hypersurface with many singularities requires more points than a regular one to obtain a good representation. As an indication, we used 10,000 points to represent the hypersurface for the Cardinal Temperature model, and 30,000 points for Ratkowsky's model.

The main problem in locating points on the hypersurface is to obtain a distribution as homogenous as possible. The main steps in our solution are as follows:
1. Random choice of a starting point at the surface of a hypersphere of radius unity with its centre at the origin.
2. Translation and scaling of this hypersphere so as to enclose the confidence region.
3. Convergence from the scaled starting point in the direction of the optimal point $\hat{\Theta}$ to obtain the coordinates of the point on the hypersurface that encloses the confidence region.
4. Projection of points belonging to the hypersurface onto different planes to obtain a graphical representation of the confidence region. These steps are summarized by Figure 2 for a model with two parameters.

1. Random Sampling

A generator of pseudo-random numbers known to behave satisfactorily for up to 100,000 successive calls (Chassé & Debouzie, 1974) was used to randomly choose a point on the surface of the hypersphere.

To obtain a homogenous distribution of points on the surface of the hypersphere, we first choose a point inside a hypercube by randomly selecting its coordinates within the range [-1,1]. When the point is outside the hypersphere of radius 1, it is rejected and a new one is chosen until it falls inside the hypersphere. When the point is inside the hypersphere, all its coordinates are divided by its distance from the origin in order to project the point on the surface of the hypersphere. A homogeneous distribution of points at the surface of the hyperspere is thereby obtained, though this solution only works for low-dimensioned spaces as the probability of obtaining a point outside the hypersphere increases dramatically with the space dimension (cf. Rubinstein, 1981). The program is therefore limited to a maximum of eight parameters.

2. Translation and Scaling

The translation of the hypersphere consists of moving its centre to the point $\hat{\Theta}$ where the SSR is minimum. Scaling consists of modifying its radius with respect to each axis to take into account scale differences for parameters. In addition, as confidence regions are sometimes highly asymmetrical with non-linear models, a different radius may be used for values below or above optimal parameter values. The scaling constants are relatively easy to choose when a biological, or graphical, interpretation of the parameter exists: for instance the parameter T_{max} in both models is approximately in the range [321-5,321+5]. However, when a parameter is meaningless this task becomes more difficult, and an iterated trial and error method to delimit its extreme values is often the sole solution.

3. Convergence to Threshold Point

In order to determine the coordinates of the point on the hypersurface enclosing the confidence region and the line which joins the starting point and $\hat{\Theta}$, we use a simple iterative method based on a linear interpolation (except that the square-root of SSR values is used to compensate for its quadratic behaviour). The stop criterion is that the relative difference between the current SSR and the threshold value must be small,

$$\frac{\text{Threshold} - \text{SSR}}{\text{Threshold}} < \varepsilon$$

where ε is a user-supplied small value, for instance 10^{-3}. The smaller ε is, the more accurate the location of the point. Furthermore, to avoid convergence failure when scaling is poor, the maximum permitted number of iterations is also a user-supplied constant, for instance 10^{+3}.

4. Graphical Representations

When the dimension of the space parameter is greater than two, a simple way to represent the hypersurface is to project it on different planes so as to obtain the shadow of the hypersurface. In general, for p parameters, $(p^2-p)/2$ projections are sufficient to represent the hypersurface.

Results

Fit to Data

Figures 3 and 4 show that both models fit the data equally well. A more sophisticated analysis would reveal that the minimum SSR for the Ratkowsky's model is three times greater than the minimum SSR for the Cardinal Temperature model and that an autocorrelation between residuals



Figure 2 Steps used to determine the confidence region parameter values, symbolized here for a model with two parameters.



Figure 3 Ratkowsky model. The best fit of the model to the data is represented in the bottom left square. The confidence region is projected on six different planes. Each cross represents a point belonging to the confidence region. 30,000 points were located for these representations. Note the high structural correlation between parameters T_{min} and b, and between parameters T_{max} and c. The marginal approximative confidence limits for parameter values (cf. Bates & Watts, 1988; Seber & Wild, 1989) have been indicated with dashed lines.

for Ratkowsky's model is visible. However, this analysis is outside the scope of this paper as the data set is clearly too poor for a biologically meaningful comparison of the two models.

The optimal parameter values obtained here for Ratkowsky's model are consistent with previously published estimates (Ratkowsky *et al.*, 1983) with 275.985 versus 276 for T_{min}, and 321.604 versus 322 for T_{max}.

Confidence Region

1. Ratkowsky's Model

A strong structural correlation is visible between parameters T_{min} and b, and between parameters T_{max} and c: the projection of the confidence region on the corresponding planes yields thin shadows (Figure 3). This structural correlation between parameters is biologically meaningless and is simply a consequence of the effect of parameters on curve patterns. Re-examination of Figure 1 shows that T_{min} and b have an opposite effect on curve patterns, so that an increase in T_{min} may be partly corrected by an increase in b, giving a thin confidence region with its main axis approximately parallel to the bisection of T_{min}-axis and b-axis. Likewise, a second look at Figure 1 shows that T_{max} and c have the same effect on curve patterns, so that an increase in T_{max} may be partly corrected by a decrease in c, giving a thin confidence region with its main axis approximately perpendicular to the bisection of the T_{max}-axis and c-axis.

Structural correlation for the determination of parameter confidence limits is tedious, since they tend to



Figure 4 Cardinal Temperature model. The best fit of the model to the data is represented in the bottom left square. The confidence region is projected on six different planes. Each cross represents a point belonging to the confidence region. 10,000 points were located for these representations. Note the absence of high structural correlation between parameters. The marginal approximative confidence limits for parameter values (cf Bates & Watts, 1988; Seber & Wild, 1989) have been indicated with dashed lines.

become large. This is clear for the biologically meaningful parameters of the Ratkowsky model: T_{max} spans over 5 K (320-325) and T_{min} over 29 K (257-286) with values below the freezing point of water, which have no biological significance. Thus, structural correlation between parameters should be avoided as it gives large confidence limits for parameter values, making comparisons difficult and extrapolations dubious. Moreover, the number of starting points required to obtain a good representation of the confidence region must be increased.

2. Cardinal Temperature Model

No high structural correlation between parameters is visible for the Cardinal Temperature model (Figure 4). The shadows of the confidence regions are always wide and regular (this is a somewhat academic example as the model was designed for this purpose). Consequently, confidence limits for parameter are acute: μ_{opt} spans over 0.19 h^{-1} (1.30-1.49), T_{min} over 6 K (286-292), T_{opt} over 2.7 K (311.8-314.4) and T_{max} over 0.8 K (320-320.8).

Conclusion

We have presented a subroutine for the determination of confidence limits for parameter values in non-linear models. The models used illustrate two extreme cases with a high positive or negative structural correlation between parameters on one hand, and the absence of high structural correlations on the other hand. The Cardinal Temperature model which was built to illustrate the latter case may be useful in itself, since all its parameters have a biological interpretation, but it must first be validated with a more extensive biological data set.

Acknowledgements

This work was supported in part by a grant from API-SYSTEM, BioMérieux, to J.R. Lobry and a grant from B.S.N. to L. Rosso.

References

BATES, D.M. & WATTS, D.G. (1988). *Non-linear Regression Analysis and its Applications*, p. 53. New York: John Wiley & Sons.

BEALE, E.M.L. (1960). Confidence regions in non-linear estimation. *Journal of the Royal Statistical Society* B **22**, 41-88.

CHASSÉ, J.L. & DEBOUZIE, D. (1974). Utilisation des test de Kiveliovitch et Vialar dans l'étude de quelques générateurs de nombres pseudo-aléatoires. *Revue de Statistique Appliquée* **22**, 83-90.

RATKOWSKY, D.A., OLLEY, J., McMEEKIN, T.A. & BALL, A. (1982). Relationship between temperature and growth rate of bacterial cultures. *Journal of Bacteriology* **149**, 1-5.

RATKOWSKY, D.A., LOWRY, R.K., McMEEKIN, T.A., STOKES, A.N. & CHANDLER, R.E. (1983). Model for bacterial culture growth rate throughout the entire biokinetic range. *Journal of Bacteriology* **154**, 1222-1226.

RUBINSTEIN, R.Y. (1981). *Simulation and the Monte Carlo Method*. New York: John Wiley & Sons.

SEBER, G.A.F. & WILD, C.J. (1989) *Non-linear Regression*, p. 192. New York: John Wiley & Sons.

Appendix

```
C------------------------------ DEMO ------------------------------
C PURPOSE
C --------
C DETERMINATION OF CONFIDENCE REGION FOR PARAMETER VALUES IN
C NON-LINEAR MODELS WITH UP TO 8 PARAMETERS.
C EXAMPLE OF APPLICATION WITH A 4-PARAMETER MODEL.
C
C VARIABLES
C ----------
C
C SSR     NAME OF AN USER-SUPPLIED SUBROUTINE WHICH RETURNS THE SUM OF
C         SQUARED RESIDUALS BETWEEN OBSERVED AND THEORETICAL VALUES.
C
C PN      NUMBER OF PARAMETERS IN THE MODEL.
C
C POPT    VECTOR WHICH CONTAINS THE PARAMETERS VALUES SUCH THAT
C         THE SSR IS MINIMUM.
C
C PINF    VECTOR WHICH CONTAINS THE PARAMETER LOWER BOUNDS TO
C         GENERATE STARTING POINTS.
C
C PSUP    VECTOR WHICH CONTAINS THE PARAMETER UPPER BOUNDS TO
C         GENERATE STARTING POINTS.
C
C THRES   THRESHOLD VALUE FOR THE SSR WHICH MAY BE OBTAINED FROM :
C
C         THRES = SSR(POPT)*(1+PN*F(alpha;PN,N-PN)/(N-PN)),
C
C          WHERE N IS THE NUMBER OF POINTS IN THE DATA SET.
C         ACCORDING TO BEALE (1960) Confidence regions in non-linear estimation.
C         Journal of the Royal Statistical Society Ser. B, 22 : 41-88.
C
C MAXPT   MAXIMUM NUMBER OF POINTS TO BE GENERATED.
C
C SEED    AN INITIAL CONDITION FOR THE PSEUDO-RANDOM NUMBERS
C         GENERATOR.
C
C FILN    THE NAME OF THE OUTPUT FILE TO WRITE COORDINATES OF POINTS.
C
C EPS     DETERMINE THE ACCURACY OF RESULTS. DURING CONVERGENCE TO THE
C         THRESHOLD, THE STOP CRITERION IS THAT (THRES-SSR)/THRES < EPS
C
C MAXSSR A STARTING POINT IS REJECTED WHEN THE RATIO OF THE SSR AT THIS
C         POINT TO THE THRESHOLD VALUE IS GREATER THAN MAXSSR:
C         (SSR/THRES > MAXSSR). THIS IS TO AVOID STARTING FROM A POINT TOO
C         FAR FROM THE OPTIMAL POINT : IN THIS CASE CONVERGENCE IS TOO
C         LONG AND SUBJECT TO FAILURE DUE TO NUMERICAL IMPRECISION.
C
C MAXIT   MAXIMUM NUMBER OF ITERATIONS ALLOWED FOR CONVERGENCE TO
C         THRESHOLD
C
C-----------------------------------------------------------------
      PROGRAM DEMO
      EXTERNAL SSR
      REAL*8 SSR,POPT(8),PINF(8),PSUP(8),THRES,SEED,EPS,MAXSSR
      INTEGER PN,MAXPT,MAXIT
      CHARACTER*30 FILN
C
C INITIALISATION OF VARIABLES :
C
      PN=4
C POPT(1) : MuOpt
      PINF(1)=1.3
      POPT(1)=1.39615
      PSUP(1)=1.5
C POPT(2) : Tmin
      PINF(2)=285.5
      POPT(2)=289.398
      PSUP(2)=292.0
C POPT(3) : Topt
      PINF(3)=311.8
      POPT(3)=313.25
      PSUP(3)=314.5
C POPT(4) : Tmax
      PINF(4)=319.9
      POPT(4)=320.234
      PSUP(4)=320.8

      THRES=0.0395811*(1+(4.0*3.36)/(15.0-4.0))
      MAXPT=10000
      SEED=0.1071966
      FILN="OUTDEMO"
      EPS=0.0001
      MAXSSR=1.0E+05
      MAXIT=500
C
C CALLING SUBROUTINE CRUNCH:
C
      CALL CRUNCH(SSR,PN,POPT,PINF,PSUP,THRES,
     &      MAXPT,SEED,FILN,EPS,MAXSSR,MAXIT)
      END
C------------------------------ SSR ------------------------------
C PURPOSE
C --------
C THIS FUNCTION RETURNS THE SUM OF SQUARED RESIDUALS BETWEEN OBSERVED
C AND THEORETICAL VALUES. FOR CLARITY, EXPERIMENTAL DATA HAVE BEEN
C INCLUDED DIRECTLY INSIDE THE PROGRAM. IT IS MORE USUAL, HOWEVER, TO
C READ DATA FROM A FILE AT THE LEVEL OF THE MAIN PROGRAM AND THEN TO
C PASS VALUES TO THE SUBROUTINE SSR VIA COMMON VARIABLES.
C
C COMPUTATION OF THE SSR IN THE CASE OF THE CARDINAL TEMPERATURE MODEL
C T1 = MuOpt, T2 = Tmin, T3 = Topt, T4 = Tmax
C
C VARIABLES
C ----------
C
C T   INPUT VECTOR WHICH CONTAINS PARAMETER VALUES.
C
C P   INPUT VARIABLE WHICH CONTAINS THE NUMBER OF PARAMETERS IN THE
C     MODEL
C
C-----------------------------------------------------------------
      REAL*8 FUNCTION SSR(T,P)
      INTEGER P
      REAL*8 T(8)
      REAL*8 A,B,T1,T2,T3,T4
      T1=T(1)
      T2=T(2)
      T3=T(3)
      T4=T(4)
      A=T2+T4
      B=T2*T4
      SSR =
     & (0.25-T1*(1-((294-T3)**2)/((294-T3)**2+294*(A-294)-B)))**2+
     & (0.56-T1*(1-((296-T3)**2)/((296-T3)**2+296*(A-296)-B)))**2+
     & (0.61-T1*(1-((298-T3)**2)/((298-T3)**2+298*(A-298)-B)))**2+
     & (0.79-T1*(1-((300-T3)**2)/((300-T3)**2+300*(A-300)-B)))**2+
     & (0.94-T1*(1-((302-T3)**2)/((302-T3)**2+302*(A-302)-B)))**2+
     & (1.04-T1*(1-((304-T3)**2)/((304-T3)**2+304*(A-304)-B)))**2+
     & (1.16-T1*(1-((306-T3)**2)/((306-T3)**2+306*(A-306)-B)))**2+
     & (1.23-T1*(1-((308-T3)**2)/((308-T3)**2+308*(A-308)-B)))**2+
     & (1.36-T1*(1-((310-T3)**2)/((310-T3)**2+310*(A-310)-B)))**2+
     & (1.32-T1*(1-((312-T3)**2)/((312-T3)**2+312*(A-312)-B)))**2+
     & (1.36-T1*(1-((314-T3)**2)/((314-T3)**2+314*(A-314)-B)))**2+
     & (1.34-T1*(1-((316-T3)**2)/((316-T3)**2+316*(A-316)-B)))**2+
     & (0.96-T1*(1-((318-T3)**2)/((318-T3)**2+318*(A-318)-B)))**2+
     & (0.83-T1*(1-((319-T3)**2)/((319-T3)**2+319*(A-319)-B)))**2+
     & (0.16-T1*(1-((320-T3)**2)/((320-T3)**2+320*(A-320)-B)))**2
      END
C------------------------------ CRUNCH ------------------------------
C PURPOSE
C --------
C THE SUBROUTINE CRUNCH WRITES COORDINATES OF POINTS SUCH THAT
C THE SSR FUNCTION EQUALS THE THRESHOLD VALUE IN THE OUTPUT FILE.
C
C-----------------------------------------------------------------
      SUBROUTINE CRUNCH(SSR,PN,POPT,PINF,PSUP,THRES,
     &        MAXPT,SEED,FILN,EPS,MAXSSR,MAXIT)
      EXTERNAL SSR
      REAL*8 SSR,POPT(8),PINF(8),PSUP(8),THRES,SEED,EPS,MAXSSR
      INTEGER PN,MAXPT,MAXIT
      CHARACTER*30 FILN
C
C DECLARATION OF LOCAL VARIABLES:
C
      REAL*8 COOR(8),TETA(8),DPINF(8),DPSUF(8),COORSV(8),TETASV(8)
      REAL*8 LOWER,UPPER,SSRLOW,SSRUP,NEW,SSRNEW
      INTEGER ITERNBR
      LOGICAL OK
C
C CHECKING POPT VALUES:
C
      IF(SSR(POPT,PN).GT.THRES)THEN
        WRITE(*,*) "ERROR FROM ROUTINE CRUNCH : SSR IS GREATER"
```

```
      WRITE(*,*) "THAN THRESHOLD WITH OPTIMAL PARAMETER VALUES"
      WRITE(*,*) "THRESHOLD = ",THRES
      WRITE(*,*) "SSR = ",SSR(POPT,PN)
      STOP
     END IF
C
C OPENING OUTPUT FILE:
C
    OPEN(UNIT=34,FILE=FILN,STATUS="NEW")
C
C INTERMEDIARY COMPUTATIONS:
C
    DO 50,I=1,PN
     DPINF(I)=(POPT(I)-PINF(I))*SQRT(DBLE(PN))
     DPSUP(I)=(PSUP(I)-POPT(I))*SQRT(DBLE(PN))
50   CONTINUE
C
C MAIN LOOP OF THE PROGRAM.
C REPETITION OF THE SAME WITH MAXPT DIFFERENT STARTING POINTS.
C
    I=1
1    I=I+1
C
C CHOOSING A POINT AT RANDOM AT THE SURFACE OF A HYPERSHERE
C OR RADIUS 1.0 WITH ITS CENTER AT THE ORIGIN:
C
     CALL SPHERE(PN,COOR,SEED)
C
C COMPUTATION OF THE CORRESPONDING PARAMETER VALUES (SCALING).
C
     DO 110,J=1,PN
      IF(COOR(J).GT.0.0)THEN
       TETA(J)=POPT(J)+COOR(J)*DPSUP(J)
      ELSE
       TETA(J)=POPT(J)+COOR(J)*DPINF(J)
      END IF
110   CONTINUE
     SSRUP=SSR(TETA,PN)
     OK=.TRUE.
C
C TESTING IF THE POINT IS OUTSIDE THE CONFIDENCE REGION:
C
     IF(SSRUP.LT.THRES)THEN
      WRITE(*,*) "WARNING 1 FROM SUBROUTINE CRUNCH"
      WRITE(*,*) "SSR LESS THAN THRESHOLD WITH :"
      WRITE(*,*) "PARAM VALUE    % DISTANCE FROM OPTIMAL PARAM VALUE"
      DO 120,K=1,PN
       WRITE(*,1000) TETA(K),100*COOR(K)*SQRT(DBLE(PN))
120    CONTINUE
      WRITE(*,*) "----------------------------------------"
      OK=.FALSE.
      I=I-1
     END IF
C
C TESTING IF SSR/THRES IS NOT TOO BIG:
C
     IF(SSRUP/THRES.GT.MAXSSR)THEN
      WRITE(*,*) "WARNING 2 FROM SUBROUTINE CRUNCH"
      WRITE(*,*) "SSR/THRES GREATER THAN : ",MAXSSR
      WRITE(*,*) "PARAM VALUE    % DISTANCE FROM OPTIMAL PARAM VALUE"
      DO 125,K=1,PN
       WRITE(*,1000) TETA(K),100*COOR(K)*SQRT(DBLE(PN))
125    CONTINUE
      WRITE(*,*) "----------------------------------------"
      OK=.FALSE.
      I=I-1
     END IF
C
C SAVING COORDINATES:
C
     DO 115,K=1,PN
      TETASV(K)=TETA(K)
      COORSV(K)=COOR(K)
115   CONTINUE
C
C CONVERGENCE TO THRESHOLD, INITIALIZATION:
C
     IF(OK)THEN
      LOWER=0.0
      UPPER=1.0
      DO 130,J=1,PN
       TETA(J)=POPT(J)
130    CONTINUE
      SSRLOW=SSR(TETA,PN)
      ITERNBR=0
C
C CONVERGENCE TO THRESHOLD, MAIN LOOP:
C
200    NEW=LOWER+(UPPER-LOWER)*SQRT((THRES-SSRLOW)/(SSRUP-SSRLOW))
      ITERNBR=ITERNBR+1
      DO 150,J=1,PN
       IF(COOR(J).GT.0.0)THEN
        TETA(J)=POPT(J)+NEW*COOR(J)*DPSUP(J)
       ELSE
        TETA(J)=POPT(J)+NEW*COOR(J)*DPINF(J)
       END IF
150    CONTINUE
      SSRNEW=SSR(TETA,PN)
      IF(SSRNEW.GT.THRES)THEN

       UPPER=NEW

          ELSE
       LOWER=NEW
      END IF
      IF(ITERNBR.LT.MAXIT)THEN
       IF(ABS(THRES-SSRNEW)/THRES.LT.EPS)THEN
        WRITE(34,1000) (TETA(K),K=1,PN)
       ELSE
        GO TO 200
       END IF
      ELSE
       WRITE(*,*) "WARNING 3 FROM SUBROUTINE CRUNCH"
       WRITE(*,*) "NUMBER OF ITERATIONS GREATER THAN",MAXIT
       WRITE(*,*) "PARAM VALUE    % DISTANCE FROM OPTIMAL PARAM VALUE"
       DO 135,K=1,PN
        WRITE(*,1000) TETASV(K),100*COORSV(K)*SQRT(DBLE(PN))
135     CONTINUE
       WRITE(*,*) "----------------------------------------"
       I=I-1
      END IF
     END IF
    IF(I.LE.MAXPT)GO TO 1
    CLOSE(34)
1000 FORMAT(' ',8(F12.6))
    END

C------------------------------ SPHERE ------------------------------
C PURPOSE
C --------
C THE SUBROUTINE SPHERE RETURNS RANDOM COORDINATES FOR A POINT
C BELONGING TO A HYPERSPHERE OF RADIUS 1.0 WITH ITS CENTER
C AT THE ORIGIN (0,0,...,0).
C
C VARIABLES
C ---------
C N        : SPACE DIMENSION : NUMBER OF COORDINATES TO GENERATE.
C            N IS IN THE RANGE [2-8]
C
C COOR     : COORDINATES OF THE POINTS RETURNED BY SPHETEST.
C
C SEED     : SEED FOR THE PSEUDO-RANDOM GENERATOR,
C            ITS VALUE HAS TO BE INITIALIZED BY THE CALLING PROGRAM
C            WITHIN [0-1] AND NOT FURTHER MODIFIED.
C
C COMMENTS
C ---------
C
C  THE N COORDINATES OF A POINT ARE CHOSEN AT RANDOM WITHIN A
C HYPERCUBE. THE DISTANCE FROM THIS POINT TO THE ORIGIN (0,0,...,0) IS
C COMPUTED. IF THE DISTANCE IS LESS THAN 1.0, THEN THE POINT IS INSIDE THE
C HYPERSPHERE, AND THE COORDINATES ARE NORMALIZED SO THAT THE
C DISTANCE TO THE ORIGIN EQUALS ONE. WHEN THE POINT IS OUTSIDE THE
C HYPERSPHERE, ANOTHER POINT IS CHOSEN AT RANDOM.
C  THIS IS TO OBTAIN A HOMOGENEOUS DISTRIBUTION OF THE
C POINTS ON THE SURFACE OF THE HYPERSPHERE.
C
C  THE OBSERVED MEAN NUMBER OF TRIALS TO OBTAIN A POINT WITHIN A
C HYPERSPHERE  IS GIVEN HERE AS FUNCTION OF THE SPACE DIMENSION N :
C
C    N    NUMBER OF TRIALS (MEAN OVER 1000)
C    2     1.284
C    3     1.970
C    4     3.243
C    5     6.010
C    6    12.193
C    7    26.699
C    8    62.023
C
C---------------------------------------------
    SUBROUTINE SPHERE(N,COOR,SEED)
    INTEGER N
    REAL*8 COOR(8),SEED,SUM

5   SUM=0.0
C
C CHOICE OF N COORDINATES AT RANDOM:
C
    DO 10,I=1,N
     COOR(I)=2.0*RANDOM(SEED)-1.0
     SUM=SUM+COOR(I)*COOR(I)
10  CONTINUE
C
C TESTING WHETHER THE POINT IS INSIDE THE HYPERSPHERE:
C
    IF(ABS(SQRT(SUM)).LE.1.0)THEN
     DO 20,I=1,N
      COOR(I)=COOR(I)/SQRT(SUM)
20    CONTINUE
    ELSE
     GO TO 5
    END IF
    END

C------------------------------ RANDOM ------------------------------
C PURPOSE
C --------
C THE FUNCTION RANDOM RETURNS A PSEUDO-RANDOM NUMBER WITHIN [0-1].
C
C VARIABLES
C ---------
C X : SEED FOR THE PSEUDO-RANDOM GENERATOR,
C    ITS VALUE HAS TO BE INITIALIZED BY THE CALLING PROGRAM
C    WITHIN [0-1] AND NOT FURTHER MODIFIED.
C
C--------------------------------------------------------------------
    FUNCTION RANDOM(X)
    REAL*8 X,RANDOM
    RANDOM=MOD(262144*3125*X,262144.0)/262144.0
    X=RANDOM

    END
```